```solidity
// SPDX-License-Identifier: GPL-3.0

pragma solidity >=0.7.0 <0.9.0;

/**
 * @title Ballot
 * @dev Implements voting process along with vote delegation
 */
contract Ballot {

    struct Voter {
        uint weight; // weight is accumulated by delegation
        bool voted;  // if true, that person already voted
        address delegate; // person delegated to
        uint vote;   // index of the voted proposal
    }

    struct Proposal {
        // If you can limit the length to a certain number of bytes,
        // always use one of bytes1 to bytes32 because they are much cheaper
        bytes32 name;   // short name (up to 32 bytes)
        uint voteCount; // number of accumulated votes
    }

    address public chairperson;

    mapping(address => Voter) public voters;

    Proposal[] public proposals;

    /**
     * @dev Create a new ballot to choose one of 'proposalNames'.
     * @param proposalNames names of proposals
     */
    constructor(bytes32[] memory proposalNames) {
        chairperson = msg.sender;
        voters[chairperson].weight = 1;

        for (uint i = 0; i < proposalNames.length; i++) {
            // 'Proposal({...})' creates a temporary
            // Proposal object and 'proposals.push(...)'
            // appends it to the end of 'proposals'.
            proposals.push(Proposal({
                name: proposalNames[i],
                voteCount: 0
            }));
```

```
47          }
48      }
49
50      /**
51       * @dev Give 'voter' the right to vote on this ballot. May only be called by 'chairperson'.
52       * @param voter address of voter
53       */
54      function giveRightToVote(address voter) public {
55          require(
56              msg.sender == chairperson,
57              "Only chairperson can give right to vote."
58          );
59          require(
60              !voters[voter].voted,
61              "The voter already voted."
62          );
63          require(voters[voter].weight == 0);
64          voters[voter].weight = 1;
65      }
66
67      /**
68       * @dev Delegate your vote to the voter 'to'.
69       * @param to address to which vote is delegated
70       */
71      function delegate(address to) public {
72          Voter storage sender = voters[msg.sender];
73          require(!sender.voted, "You already voted.");
74          require(to != msg.sender, "Self-delegation is disallowed.");
75
76          while (voters[to].delegate != address(0)) {
77              to = voters[to].delegate;
78
79              // We found a loop in the delegation, not allowed.
80              require(to != msg.sender, "Found loop in delegation.");
81          }
82          sender.voted = true;
83          sender.delegate = to;
84          Voter storage delegate_ = voters[to];
85          if (delegate_.voted) {
86              // If the delegate already voted,
87              // directly add to the number of votes
88              proposals[delegate_.vote].voteCount += sender.weight;
89          } else {
90              // If the delegate did not vote yet,
91              // add to her weight.
92              delegate_.weight += sender.weight;
```

```
        }
    }

    /**
     * @dev Give your vote (including votes delegated to you) to proposal 'proposals[proposal].name'.
     * @param proposal index of proposal in the proposals array
     */
    function vote(uint proposal) public {
        Voter storage sender = voters[msg.sender];
        require(sender.weight != 0, "Has no right to vote");
        require(!sender.voted, "Already voted.");
        sender.voted = true;
        sender.vote = proposal;

        // If 'proposal' is out of the range of the array,
        // this will throw automatically and revert all
        // changes.
        proposals[proposal].voteCount += sender.weight;
    }

    /**
     * @dev Computes the winning proposal taking all previous votes into account.
     * @return winningProposal_ index of winning proposal in the proposals array
     */
    function winningProposal() public view
            returns (uint winningProposal_)
    {
        uint winningVoteCount = 0;
        for (uint p = 0; p < proposals.length; p++) {
            if (proposals[p].voteCount > winningVoteCount) {
                winningVoteCount = proposals[p].voteCount;
                winningProposal_ = p;
            }
        }
    }

    /**
     * @dev Calls winningProposal() function to get the index of the winner contained in the proposals array and then
     * @return winnerName_ the name of the winner
     */
    function winnerName() public view
            returns (bytes32 winnerName_)
    {
        winnerName_ = proposals[winningProposal()].name;
    }
}
```